Example 2
Splitting Functions in QCD

Now, we are going to calculate a leading-order Pgq splitting function in massless QCD from electron-positron annihilation process.

```
SetDirectory[NotebookDirectory[]];
<< LiteRed`
```

```
**************** LiteRed v1.82 ********************
Author: Roman N. Lee, Budker Institute of Nuclear Physics, Novosibirsk.
Release Date: 01.06.2015
LiteRed stands for Loop InTEgrals REDuction.
The package is designed for the search and application of the
   Integration-By-Part reduction rules. It also contains some other useful tools.
See ?LiteRed`* for a list of functions.
```

1. Physical Setup

We consider the process $e^{+}(q_1) + e^{-}(q_2) \rightarrow q(p_1) +$ qbar $(p_2) + g(p_3)$
The criss-section for this process is given by

$$\text{M2} = \frac{\text{sp[p1, q1]}^2 + \text{sp[p1, q2]}^2 + \text{sp[p2, q1]}^2 + \text{sp[p2, q2]}^2}{\text{sp[q1, q2] sp[p1, p3] sp[p2, p3]}};$$

2. Integration-by-Parts

Initializaion

```
SetDim[n];
Declare[{x, s}, Number, {q1, q2, p1, p2, p3}, Vector];
sp[q1] = 0;
sp[q2] = 0;
sp[q1, q2] = s / 2;
```

Define Basis #1

```
NewBasis[$a, {sp[p1], sp[p3], sp[q1 + q2 - p1 - p3],
   s x - 2 sp[q1 + q2, p3], sp[p1, p3], sp[q1 + q2 - p1, p3]},
  {p1, p3}, Append → True, GenerateIBP → True, Directory → "b1.ibp"]
```

NewBasis::notb : Overdetermined set of denominators. Not a basis. Aborting...

```
$Aborted
```

Partial Fractioning

Taking into account that $\frac{1}{\text{sp[p1,p3] sp[p2,p3]}} = \frac{1}{2x}\left(\frac{1}{\text{sp}[p_1,p_3]} + \frac{1}{\text{sp}[p_2,p_3]}\right)$where $x = \frac{\text{sp}[q_1+q_2,p_3]}{\text{sp}[q_1,q_2]}$we split M2=M2a+M2b like this

```
M2a = (sp[p1, q1]^2 + sp[p1, q2]^2 + sp[p2, q1]^2 + sp[p2, q2]^2) / (2 x sp[q1, q2] sp[p1, p3]);
M2b = (sp[p1, q1]^2 + sp[p1, q2]^2 + sp[p2, q1]^2 + sp[p2, q2]^2) / (2 x sp[q1, q2] sp[p2, p3]);
```

Since these contributions are symmetrical (we can replace p_1 and p_2) we will consider only one of them, let it be M2a.

Define Basis #2

```
NewBasis[$b, {sp[p1], sp[p3], sp[q1 + q2 - p1 - p3], s x - 2 sp[q1 + q2, p3], sp[p1, p3]},
  {p1, p3}, Append → True, GenerateIBP → True, Directory → "b1.ibp"]
```

```
Irreducible numerator(s) appended: p1 · q1,p3 · q1.
Pattern for AnalyzeSectors: {___,0, 0}

Valid basis.
    Ds[$b] — denominators,
    SPs[$b] — scalar products involving loop momenta,
    LMs[$b] — loop momenta,
    EMs[$b] — external momenta,
    Toj[$b] — rules to transform scalar products to denominators.
The definitions of the basis will be saved in b1.ibp

Integration-By-Part&Lorentz-Invariance identities are generated.
    IBP[$b] — integration-by-part identities,
    LI[$b] — Lorentz invariance identities.
```

```
AnalyzeSectors[$b, {___, 0, 0}, CutDs → {1, 1, 1, 1, 0, 0, 0}];
```

```
Found 30 zero sectors out of 32.
    ZeroSectors[$b] — zero sectors,
    NonZeroSectors[$b] — nonzero sectors,
    SimpleSectors[$b] — simple sectors (no nonzero subsectors),
    BasisSectors[$b] — basis sectors (at least one immediate subsector is zero),
    ZerojRule[$b] — a rule to nullify all zero j[$b...],
    CutDs[$b] — a flag list of cut denominatorsj (1=cut).
```

```
FindSymmetries[$b]
```

```
Found 0 mapped sectors and 2 unique sectors.
    UniqueSectors[$b] — unique sectors.
    MappedSectors[$b] — mapped sectors.
    SR[$b][…] — symmetry relations for j[$b,…] from UniqueSectors[$b].
    jSymmetries[$b,…] — symmetry rules for the sector js[$b,…] in UniqueSectors[$b].
    jRules[$b,…] — reduction rules for j[$b,…] from MappedSectors[$b].
```

```
SolvejSector /@ UniqueSectors[$b]
```

```
Sector js[$b, 1, 1, 1, 1, 0, 0, 0]

    1 master integrals found:
j[$b, 1, 1, 1, 1, 0, 0, 0].
    jRules[$b, 1, 1, 1, 1, 0, 0, 0] — reduction rules for the sector.
    MIs[$b] — updated list of the masters.

Sector js[$b, 1, 1, 1, 1, 1, 0, 0]

    0 master integrals found.
    jRules[$b, 1, 1, 1, 1, 1, 0, 0] — reduction rules for the sector.
    MIs[$b] — updated list of the masters.

{1, 0}
```

Master Integrals

```
MIs[$b]
% // Fromj
```

```
{j[$b, 1, 1, 1, 1, 0, 0, 0]}

{1 / (p1 · p1 p3 · p3 (s x - 2 p3 · (q1 + q2)) (-p1 - p3 + q1 + q2) · (-p1 - p3 + q1 + q2))}
```

3. Integration of the Cross-Section

Phase-space “measure”:

```
$PS2 = 1 / (sp[p1] sp[p3] sp[q1 + q2 - p1 - p3] (s x - 2 sp[q1 + q2, p3]));
```

```
$M2 = M2a /. p2 → q1 + q2 - p1 - p3
```

$$\frac{1}{s\,x\,p1\cdot p3}\left((p1\cdot q1)^2+(p1\cdot q2)^2+(q1\cdot(-p1-p3+q1+q2))^2+(q2\cdot(-p1-p3+q1+q2))^2\right)$$

```
$jM2 = Toj[$b, $PS2 $M2]
```

$$
\begin{aligned}
&\frac{j[\$b,-1,1,1,1,1,0,0]}{2\,s\,x} + \frac{j[\$b,0,0,1,1,1,0,0]}{s\,x} - \frac{j[\$b,0,1,0,1,1,0,0]}{s\,x} + \\
&\frac{j[\$b,0,1,1,0,1,0,0]}{2\,s\,x} + \frac{2\,j[\$b,0,1,1,1,0,0,0]}{s\,x} - \frac{2\,j[\$b,0,1,1,1,1,-1,0]}{s\,x} - \\
&\frac{j[\$b,0,1,1,1,1,0,-1]}{s\,x} - \frac{(-1+x)\,j[\$b,0,1,1,1,1,0,0]}{2\,x} + \\
&\frac{j[\$b,1,-1,1,1,1,0,0]}{2\,s\,x} - \frac{j[\$b,1,0,0,1,1,0,0]}{s\,x} + \frac{j[\$b,1,0,1,0,1,0,0]}{2\,s\,x} + \\
&\frac{2\,j[\$b,1,0,1,1,0,0,0]}{s\,x} - \frac{2\,j[\$b,1,0,1,1,1,-1,0]}{s\,x} - \frac{j[\$b,1,0,1,1,1,0,-1]}{s\,x} - \\
&\frac{(-1+x)\,j[\$b,1,0,1,1,1,0,0]}{2\,x} + \frac{j[\$b,1,1,-1,1,1,0,0]}{2\,s\,x} - \\
&\frac{j[\$b,1,1,0,0,1,0,0]}{2\,s\,x} - \frac{2\,j[\$b,1,1,0,1,0,0,0]}{s\,x} + \frac{2\,j[\$b,1,1,0,1,1,-1,0]}{s\,x} + \\
&\frac{j[\$b,1,1,0,1,1,0,-1]}{s\,x} + \frac{(-1+x)\,j[\$b,1,1,0,1,1,0,0]}{2\,x} + \\
&\frac{j[\$b,1,1,1,-1,1,0,0]}{4\,s\,x} + \frac{j[\$b,1,1,1,0,0,0,0]}{s\,x} - \\
&\frac{j[\$b,1,1,1,0,1,-1,0]}{s\,x} - \frac{(-1+x)\,j[\$b,1,1,1,0,1,0,0]}{2\,x} + \\
&\frac{2\,j[\$b,1,1,1,1,-1,0,0]}{s\,x} - \frac{4\,j[\$b,1,1,1,1,0,-1,0]}{s\,x} - \\
&\frac{2\,j[\$b,1,1,1,1,0,0,-1]}{s\,x} - \frac{(-1+x)\,j[\$b,1,1,1,1,0,0,0]}{x} + \\
&\frac{4\,j[\$b,1,1,1,1,1,-2,0]}{s\,x} + \frac{4\,j[\$b,1,1,1,1,1,-1,-1]}{s\,x} + \\
&\frac{(-2+x)\,j[\$b,1,1,1,1,1,-1,0]}{x} + \frac{2\,j[\$b,1,1,1,1,1,0,-2]}{s\,x} - \\
&\frac{j[\$b,1,1,1,1,1,0,-1]}{x} + \frac{s\,(2-2\,x+x^2)\,j[\$b,1,1,1,1,1,0,0]}{4\,x}
\end{aligned}
$$

```
$jM2 = $jM2 // IBPReduce
```

$$
\left(n\left(-12+4\,n+12\,x-4\,n\,x-2\,x^2+n\,x^2\right) j[\$b,1,1,1,1,0,0,0]\right) \Big/ \left(4\,(-4+n)\,(-1+n)\,x^2\right)
$$

```
$Pgq = Series[$jM2 /. {n → 4 - 2 eps}, {eps, 0, -1}] // Simplify
```

$$
-\frac{(2-2\,x+x^2)\,j[\$b,1,1,1,1,0,0,0]}{3\,x^2\,\text{eps}} + O[\text{eps}]^0
$$

4. Differential Equations

Derive ODE

Master Integral

```
$mi = j[$b, 1, 1, 1, 1, 0, 0, 0];
```

```
$dmi = Toj[$b, D[Fromj[$mi], x]] // IBPReduce
$ode = {{$dmi /. {n → 4 - 2 eps, j[__] → 1}}} // Simplify
```

$$\frac{(6-2\,n-10\,x+3\,n\,x)\;j[\$b,1,1,1,1,0,0,0]}{2\,(-1+x)\,x}$$

$$\left\{\left\{\frac{-1+eps\,(2-3\,x)+x}{(-1+x)\,x}\right\}\right\}$$

```
Put[$ode, "ex4_ode.m"]
```

Find eps-form with Fuchsia

```
$ode$eps = Get["ex4_ode_eps.m"]
```

$$\left\{\left\{-\frac{eps}{-1+x}-\frac{2\,eps}{x}\right\}\right\}$$

```
$t$eps = Get["ex4_ode_eps_t.m"]
```

$$\{\{-x\}\}$$

Solution

At the leading order we can say that our master integral is proportional to the transformation matrix. In other words:

```
$result = $Pgq /. $mi → C x
```

$$-\frac{C\,(2-2\,x+x^2)}{3\,x\,eps}+O[eps]^0$$

5. Final Result

Finally, we have found a gluon-quark splitting function up to some normalization constatn:

```
Pgq = - (3 eps / C) $result // Normal
```

$$\frac{2 - 2\,x + x^2}{x}$$