Example #1
Massive One-Loop Propagator

Initialization

```
SetDirectory[NotebookDirectory[]];
<< LiteRed`
```

```
**************** LiteRed v1.82 ********************
Author: Roman N. Lee, Budker Institute of Nuclear Physics, Novosibirsk.
Release Date: 01.06.2015
LiteRed stands for Loop InTEgrals REDuction.
The package is designed for the search and application of the
   Integration-By-Part reduction rules. It also contains some other useful tools.
See ?LiteRed`* for a list of functions.
```

1. Integration-by-Parts

Initialization

```
SetDim[n];
Declare[{m2}, Number, {l, p}, Vector];
```

Define Basis

```
NewBasis[$b, {sp[l] - m2, sp[l - p] - m2}, {l}, Directory → "ex1.ibp", GenerateIBP → True]
```

```
Valid basis.
    Ds[$b] — denominators,
    SPs[$b] — scalar products involving loop momenta,
    LMs[$b] — loop momenta,
    EMs[$b] — external momenta,
    Toj[$b] — rules to transform scalar products to denominators.
The definitions of the basis will be saved in ex1.ibp
```

DiskSave::overwrite : The file ex1.ibp/$b has been overwritten.

```
Integration-By-Part&Lorentz-Invariance identities are generated.
    IBP[$b] — integration-by-part identities,
    LI[$b] — Lorentz invariance identities.
```

Examples

Already at this point we can perform some useful operations.

Fromj

```
j[$b, 3, -1] // Fromj
```

$$\frac{-m2 + (l-p)\cdot(l-p)}{(-m2 + l\cdot l)^3}$$

Toj

```
Toj[$b, sp[l, p]]
```

$$\frac{1}{2}\,j[\$b, -1, 0] - \frac{1}{2}\,j[\$b, 0, -1] + \frac{1}{2}\,j[\$b, 0, 0]\; p\cdot p$$

What follows is an example of **WRONG usage** of the basis **$b**.
Functions j's should not appear in the denominator.
In this case we need to define a new basis which explicitely contains `sp[l,p]`.

```
Toj[$b, 1 / sp[l, p]]
```

$$\frac{2\,j[\$b, 0, 0]}{j[\$b, -1, 0] - j[\$b, 0, -1] + p\cdot p}$$

Analyze Sectors

```
AnalyzeSectors[$b]
```

```
Found 1 zero sectors out of 4.
    ZeroSectors[$b] — zero sectors,
    NonZeroSectors[$b] — nonzero sectors,
    SimpleSectors[$b] — simple sectors (no nonzero subsectors),
    BasisSectors[$b] — basis sectors (at least one immediate subsector is zero),
    ZerojRule[$b] — a rule to nullify all zero j[$b...],
    CutDs[$b] — a flag list of cut denominatorsj (1=cut).
```

Find Symmetries

```
FindSymmetries[$b]
```

```
Found 1 mapped sectors and 2 unique sectors.
    UniqueSectors[$b] — unique sectors.
    MappedSectors[$b] — mapped sectors.
    SR[$b][...] — symmetry relations for j[$b,...] from UniqueSectors[$b].
    jSymmetries[$b,...] — symmetry rules for the sector js[$b,...] in UniqueSectors[$b].
    jRules[$b,...] — reduction rules for j[$b,...] from MappedSectors[$b].
```

Find IBP Rules

```
UniqueSectors[$b]
SolvejSector /@ %
```

```
{js[$b, 0, 1], js[$b, 1, 1]}
```

```
Sector js[$b, 0, 1]
```

DiskSave::overwrite : The file ex1.ibp/jRules[$b, 0, 1] has been overwritten.

```
    1 master integrals found:
j[$b, 0, 1].
    jRules[$b, 0, 1] — reduction rules for the sector.
    MIs[$b] — updated list of the masters.
```

```
Sector js[$b, 1, 1]
```

DiskSave::overwrite : The file ex1.ibp/jRules[$b, 1, 1] has been overwritten.

```
    1 master integrals found:
j[$b, 1, 1].
    jRules[$b, 1, 1] — reduction rules for the sector.
    MIs[$b] — updated list of the masters.
```

```
{1, 1}
```

Master Integrals

```
MIs[$b]
Fromj /@ %
```

```
{j[$b, 0, 1], j[$b, 1, 1]}
```

$$\left\{\frac{1}{-\mathrm{m2}+(\mathrm{l}-\mathrm{p})\cdot(\mathrm{l}-\mathrm{p})},\ \frac{1}{(-\mathrm{m2}+\mathrm{l}\cdot\mathrm{l})\,(-\mathrm{m2}+(\mathrm{l}-\mathrm{p})\cdot(\mathrm{l}-\mathrm{p}))}\right\}$$

Examples

2. Differential Equations

Definition

```
sp[p] = p2;
$dmis = IBPReduce[Dinv[#, sp[p, p]]] & /@ MIs[$b]
```

$$\left\{0,\ \frac{(-2+n)\,\mathrm{j}[\$b,0,1]}{(4\,\mathrm{m2}-\mathrm{p2})\,\mathrm{p2}}-\frac{(4\,\mathrm{m2}-4\,\mathrm{p2}+n\,\mathrm{p2})\,\mathrm{j}[\$b,1,1]}{2\,(4\,\mathrm{m2}-\mathrm{p2})\,\mathrm{p2}}\right\}$$

```
$ode = Coefficient[#, $jmis] & /@ $dmis
$ode = $ode //. {n → 4 - 2 eps} // Simplify;
```

$$\left\{\{0,\ 0\},\ \left\{\frac{-2+n}{(4\,m2-p2)\ p2},\ -\frac{4\,m2-4\,p2+n\,p2}{2\ (4\,m2-p2)\ p2}\right\}\right\}$$

ϵ-form

To find the epsilon form we are going to use Fuchsia.
For thay we save our initial equations to the file, run Fuchsia from the shell, and read back the espilon form.

```
Put[$ode, "ex1_ode.m"]
```

Let us read the epsilon form and the corresponding transformation generated by Fuchsia.

```
$ode$eps = Get["ex1_ode_y_eps.m"];
$ode$eps // MatrixForm
```

$$\begin{pmatrix} 0 & 0 \\ -\frac{2\,\mathrm{eps}}{3\,m2\,(-1+y)}+\frac{2\,\mathrm{eps}}{3\,m2\,(1+y)} & \frac{\mathrm{eps}}{-1+y}+\frac{\mathrm{eps}}{1+y} \end{pmatrix}$$

```
$ode$t = Get["ex1_ode_y_eps_t.m"];
$ode$t // MatrixForm
```

$$\begin{pmatrix} \frac{4\,(-1+2\,\mathrm{eps})}{3\,(-1+\mathrm{eps})} & 0 \\ \frac{4}{3\,m2} & -\frac{2}{y} \end{pmatrix}$$

Solution

ODE Solver

```
$SolveODE[m_, x_, ep_, n_, c_] := Module[
   {$i, $j, $n, $sol, $sol0, $sol1},

   $n = Length[m];
   $sol[0] = Table[c[$j, 0], {$j, 1, $n}];
   For[$i = 1, $i ≤ n, $i++,
    $sol0 = Table[c[$j, $i], {$j, 1, $n}];
    $sol1 = Integrate[Dot[#, $sol[$i - 1]] // Expand, x] & /@ m;
    $sol[$i] = $sol0 + $sol1;
   ];
   Sum[ep^$i * $sol[$i], {$i, 0, n}]
  ];
```

Solution for the ϵ-form

```
$mis$eps = $SolveODE[$ode$eps/eps, y, eps, 1, C]
```

$$\left\{C[1,0] + eps\, C[1,1],\ C[2,0] + eps\left(C[2,1] + \frac{1}{3\,m2}((-2\,C[1,0] + 3\,m2\,C[2,0])\,Log[1-y] + (2\,C[1,0] + 3\,m2\,C[2,0])\,Log[1+y])\right)\right\}$$

Solution for the Initial Masters

```
$mis = Dot[$ode$t, $mis$eps];
$mis = Series[$mis, {eps, 0, 1}] // Simplify
```

$$\left\{\frac{4}{3}C[1,0] - \frac{4}{3}(C[1,0] - C[1,1])\,eps + O[eps]^2,\ \left(\frac{4\,C[1,0]}{3\,m2} - \frac{2\,C[2,0]}{y}\right) + \frac{1}{3\,m2\,y}(4\,y\,C[1,1] - 6\,m2\,C[2,1] + (4\,C[1,0] - 6\,m2\,C[2,0])\,Log[1-y] - 2\,(2\,C[1,0] + 3\,m2\,C[2,0])\,Log[1+y])\,eps + O[eps]^2\right\}$$

```
$mis$rules = MapThread[#1 → #2 &, {MIs[$b], $mis}]
```

$$\left\{j[\$b,0,1] \to \frac{4}{3}C[1,0] - \frac{4}{3}(C[1,0] - C[1,1])\,eps + O[eps]^2,\ j[\$b,1,1] \to \left(\frac{4\,C[1,0]}{3\,m2} - \frac{2\,C[2,0]}{y}\right) + \frac{1}{3\,m2\,y}(4\,y\,C[1,1] - 6\,m2\,C[2,1] + (4\,C[1,0] - 6\,m2\,C[2,0])\,Log[1-y] - 2\,(2\,C[1,0] + 3\,m2\,C[2,0])\,Log[1+y])\,eps + O[eps]^2\right\}$$

Constants of Integration

Master #1

Our Solution

```
$mi1 = $mis[[1]] // Simplify
```

$$\frac{4}{3}C[1,0] - \frac{4}{3}(C[1,0] - C[1,1])\,eps + O[eps]^2$$

Smirnov's Solution

From Smirnov's book.

```
F1[a_, m2_] := (-1)^a Gamma[a - 2 + eps]/Gamma[a] m2^(2-eps-a)
```

$$F1[a\_, m2\_] := (-1)^a \frac{\text{Gamma}[a-2+\text{eps}]}{\text{Gamma}[a]} m2^{2-\text{eps}-a}$$

```
F1[1, m2] // Series[#, {eps, 0, 0}] &
```

$$\frac{m2}{\text{eps}} + (m2 - \text{EulerGamma}\ m2 - m2\ \text{Log}[m2]) + O[\text{eps}]^1$$

Integration Constants

$$\$C1 = \left\{C[1,0] \to \frac{3\ m2}{4\ \text{eps}},\ C[1,1] \to \frac{3}{4}\ \frac{2\ m2 - \text{EulerGamma}\ m2 - m2\ \text{Log}[m2]}{\text{eps}}\right\};$$

Check #1

```
$mi1 /. $C1
```

$$\frac{m2}{\text{eps}} + (m2 - \text{EulerGamma}\ m2 - m2\ \text{Log}[m2]) + O[\text{eps}]^2$$

a=13

Smirnov's

```
F1[13, m2] // Series[#, {eps, 0, 1}] &
```

$$-\frac{1}{132\ m2^{11}} + \frac{(-7381 + 2520\ \text{EulerGamma} + 2520\ \text{Log}[m2])\ \text{eps}}{332\,640\ m2^{11}} + O[\text{eps}]^2$$

Our

```
j[$b, 0, 13] // IBPReduce
(% /. $mis$rules /. $C1 /. n → 4 - 2 eps) // Series[# // Normal, {eps, 0, 1}] &
```

$$((-24+n)\ (-22+n)\ (-20+n)\ (-18+n)\ (-16+n)\ (-14+n)\ (-12+n)\ (-10+n)\ (-8+n)\ (-6+n)\ (-4+n)\ (-2+n)\ j[\$b, 0, 1]) \Big/ \left(1\,961\,990\,553\,600\ m2^{12}\right)$$

$$-\frac{1}{132\ m2^{11}} + \frac{(-7381 + 2520\ \text{EulerGamma} + 2520\ \text{Log}[m2])\ \text{eps}}{332\,640\ m2^{11}} + O[\text{eps}]^2$$

Master #2

Our Solution

```
$mis[[2]]
$mi2 = Series[% /. $C1 // Normal, {eps, 0, 1}]
```

$$\left(\frac{4\,C[1,0]}{3\,m2}-\frac{2\,C[2,0]}{y}\right)+\frac{1}{3\,m2\,y}\Big(4\,y\,C[1,1]-6\,m2\,C[2,1]+(4\,C[1,0]-6\,m2\,C[2,0])\,\mathrm{Log}[1-y]-2\,(2\,C[1,0]+3\,m2\,C[2,0])\,\mathrm{Log}[1+y]\Big)\,eps+O[eps]^2$$

$$\frac{1}{eps}+\frac{1}{y}(2\,y-\mathrm{EulerGamma}\,y-2\,C[2,0]-y\,\mathrm{Log}[m2]+\mathrm{Log}[1-y]-\mathrm{Log}[1+y])-\frac{1}{y}2\,(C[2,1]+C[2,0]\,\mathrm{Log}[1-y]+C[2,0]\,\mathrm{Log}[1+y])\,eps+O[eps]^2$$

Integration Constants

```
$C2 = {C[2, 0] → 0}
```

$\{C[2,0]\to 0\}$

```
$mi2 /. $C1 /. $C2 // Normal // Simplify
```

$$2+\frac{1}{eps}-\mathrm{EulerGamma}-\frac{2\,eps\,C[2,1]}{y}-\mathrm{Log}[m2]+\frac{\mathrm{Log}[1-y]}{y}-\frac{\mathrm{Log}[1+y]}{y}$$